
Values, and Wholemeal
Programming



Course Materials and Tools

• Course homepage:
http://flolac.iis.sinica.edu.tw/flolac18/

• We will be using the Glasgow Haskell Compiler (GHC).
• A Haskell compiler written in Haskell, with an interpreter
that both interprets and runs compiled code.

• Installation: the Haskell Platform:
http://hackage.haskell.org/platform/

• Early parts of the course material are adapted from Bird,
which I highly recommend.

Function Definition

• A function definition consists of a type declaration, and
the definition of its body:

square :: Int→ Int
square x = x× x

smaller :: Int→ Int→ Int
smaller x y = if x ≤ y then x else y

• The GHCi interpreter evaluates expressions in the loaded
context:

? square 3768
14197824
? square (smaller 5 (3+ 4))
25

Evaluation

One possible sequence of evaluating (simplifying, or reducing)
square (3+ 4):

square (3+ 4)

= { definition of + }
square 7

= { definition of square }
7× 7

= { definition of × }
49

Another Evaluation Sequence

• Another possible reduction sequence:
square (3+ 4)

= { definition of square }
(3+ 4)× (3+ 4)

= { definition of + }
7× (3+ 4)

= { definition of + }
7× 7

= { definition of × }
49

• In this sequence the rule for square is applied first. The
final result stays the same.

• Do different evaluations orders always yield the same
thing?

A Non-terminating Reduction

• Consider the following program:
three :: Int→ Int
three x = 3
infinity :: Int
infinity = infinity+ 1

• Try evaluating three infinity. If we simplify infinity first:
three infinity

= { definition of infinity }
three (infinity+ 1)

= three ((infinity+ 1) + 1) . . .
• If we start with simplifying three:

three infinity
= { definition of three }

3

Evaluation Order

• There can be many other evaluation orders. As we have
seen, some terminates while some do not.

• normal form: an expression that cannot be reduced
anymore.

• 49 is in normal form, while 7× 7 is not.
• Some expressions do not have a normal form. E.g. infinity.

• A corollary of the Church–Rosser theorem: an expression
has at most one normal form.

• If two evaluation sequences both terminate, they reach the
same normal form.

Evaluation Order

• Applicative order evaluation: starting with the innermost
reducible expression (a redex).

• Normal order evaluation: starting with the outermost
redex.

• If an expression has a normal form, normal order
evaluation delivers it. Hence the name.

• For now you can imagine that Haskell uses normal order
evaluation. A way to implement normal order evaluation
is called lazy evaluation.

Forced Computation

How to evaluate positive (3+ 4)?

positive 0 = 1
positive n = n× n .

There is only one way:
positive (3+ 4)

= positive 7
= 7× 7 = 49 .

Huh? Shouldn’t the outermost identifier be expanded and
reduced in normal order evaluation?

Forced Computation

In this case, the outermost redex is not postive, but 3+ 4.

An inner sub-expression is the redex when we need to
examine its value to determine how to carry on.

This could happen at the site of

• pattern matching,
• guarded expressions,
• case expressions,
• some built-in functions such as (<), (≤), etc...

Booleans

The datatype Bool can be introduced with a datatype
declaration:

data Bool = False | True

(But you need not do so. The type Bool is already defined in
the Haskell Prelude.)

Datatype Declaration

• In Haskell, a data declaration defines a new type.
data Type = Con1 Type11 Type12 . . .

| Con2 Type21 Type22 . . .
| :

• The declaration above introduces a new type, Type, with
several cases.

• Each case starts with a constructor, and several (zero or
more) arguments (also types).

• Informally it means “a value of type Type is either a Con1
with arguments Type11, Type12…, or a Con2 with arguments
Type21, Type22…”

• Types and constructors begin in capital letters.

12 / 214



Functions on Booleans

Negation:

not :: Bool→ Bool
not False = True
not True = False

• Notice the definition by pattern matching. The definition
has two cases, because Bool is defined by two cases. The
shape of the function follows the shape of its argument.

Functions on Booleans

Conjunction and disjunction:

(∧), (∨) :: Bool→ Bool→ Bool
False ∧ x = False
True ∧ x = x
False ∨ x = x
True ∨ x = True

I use the symbols ∧ and ∨ due to mathematical convension. In
your Haskell code, ∧ should be written &&, and ∨ should be ||.

14 / 214



Functions on Booleans

Equality check:

(= =), (̸=) :: Bool→ Bool→ Bool
x = = y = (x ∧ y) ∨ (not x ∧ not y)
x ̸= y = not (x = = y)

• = is a definition, while == is a function.
• = = and ̸= are written respectively written == and / = in
ASCII.

Characters

• You can think of Char as a big data definition:

data Char = ’a’ | ’b’ | . . .

with functions:
ord :: Char→ Int
chr :: Int→ Char

• Characters are compared by their order:

isDigit :: Char→ Bool
isDigit x = ’0’ ≤ x ∧ x ≤ ’9’

Equality Check

• Of course, you can test equality of characters too:

(= =) :: Char→ Char→ Bool

• (= =) is an overloaded name — one name shared by many
different definitions of equalities, for different types:

• (= =) :: Int→ Int→ Bool
• (= =) :: (Int, Char)→ (Int, Char)→ Bool
• (= =) :: [Int]→ [Int]→ Bool ...

• Haskell deals with overloading by a general mechanism
called type classes. It is considered a major feature of
Haskell.

• While the type class is an interesting topic, we might not
cover much of it since it is orthogonal to the central
message of this course.

Tuples

• The polymorphic type (a,b) is essentially the same as the
following declaration:

data Pair a b = MkPair a b

• Or, had Haskell allow us to use symbols:

data (a,b) = (a,b)

• Two projections:

fst :: (a,b)→ a
fst (a,b) = a
snd :: (a,b)→ b
snd (a,b) = b

Lists in Haskell

• Traditionally an important datatype in functional
languages.

• In Haskell, all elements in a list must be of the same type.
• [1, 2, 3, 4] :: [Int]
• [True, False, True] :: [Bool]
• [[1, 2], [], [6, 7]] :: [[Int]]
• [] :: List a, the empty list (whose element type is not
determined).

• String is an abbreviation for [Char]; ”abcd” is an
abbreviation of [’a’,’b’,’c’,’d’].

List as a Datatype

• [] :: List a is the empty list whose element type is not
determined.

• If a list is non-empty, the leftmost element is called its
head and the rest its tail.

• The constructor (:) :: a→ List a→ List a builds a list. E.g.
in x : xs, x is the head and xs the tail of the new list.

• You can think of a list as being defined by

data List a = [] | a : List a

• [1, 2, 3] is an abbreviation of 1 : (2 : (3 : [])).

Head and Tail

• head :: List a→ a. e.g. head [1, 2, 3] = 1.
• tail :: List a→ List a. e.g. tail [1, 2, 3] = [2, 3].
• init :: List a→ List a. e.g. init [1, 2, 3] = [1, 2].
• last :: List a→ a. e.g. last [1, 2, 3] = 3.
• They are all partial functions on non-empty lists. e.g.
head [] = ⊥.

• null :: List a→ Bool checks whether a list is empty.

List Generation

• [0..25] generates the list [0, 1, 2..25].
• [0, 2..25] yields [0, 2, 4..24].
• [2..0] yields [].
• The same works for all ordered types. For example Char:

• [’a’..’z’] yields [’a’,’b’,’c’..’z’].
• [1..] yields the infinite list [1, 2, 3..].

22 / 214



List Comprehension

• Some functional languages provide a convenient notation
for list generation. It can be defined in terms of simpler
functions.

• e.g. [x× x | x← [1..5],odd x] = [1, 9, 25].
• Syntax: [e | Q1,Q2..]. Each Qi is either

• a generator x← xs, where x is a (local) variable or pattern
of type a while xs is an expression yielding a list of type
List a, or

• a guard, a boolean valued expression (e.g. odd x).
• e is an expression that can involve new local variables
introduced by the generators.

List Comprehension

Examples:

• [(a,b) | a← [1..3],b← [1..2]] =
[(1, 1), (1, 2), (2, 1), (2, 2), (3, 1), (3, 2)]

• [(a,b) | b← [1..2],a← [1..3]] =
[(1, 1), (2, 1), (3, 1), (1, 2), (2, 2), (3, 2)]

• [(i, j) | i← [1..4], j← [i+ 1..4]] =
[(1, 2), (1, 3), (1, 4), (2, 3), (2, 4), (3, 4)]

• [(i, j) |← [1..4], even i, j← [i+ 1..4],odd j] = [(2, 3)]

Two Modes of Programming

• Functional programmers switch between two modes of
programming.

• Inductive/recursive mode: go into the structure of the
input data and recursively process it.

• Combinatorial mode: compose programs using existing
functions (combinators), process the input in stages.

• We will try the latter style today. However, that means we
have to familiarise ourselves to a large collection of library
functions.

• In the next lecture we will talk about how these library
functions can be defined, in the former style.

Length and Indexing

• (!!) :: List a→ Int→ a. List indexing starts from zero. e.g.
[1, 2, 3]!!0 = 1.

• length :: List a→ Int. e.g. length [0..9] = 10.

Append and Concatenation

• Append: (++) :: List a→ List a→ List a. In ASCII one types
(++).

• [1, 2] ++[3, 4, 5] = [1, 2, 3, 4, 5]
• [] ++[3, 4, 5] = [3, 4, 5] = [3, 4, 5] ++[]

• Compare with (:) :: a→ List a→ List a. It is a type error to
write [] : [3, 4, 5]. (++) is defined in terms of (:).

• concat :: List (List a)→ List a.
• e.g. concat [[1, 2], [], [3, 4], [5]] = [1, 2, 3, 4, 5].
• concat is defined in terms of (++).

Take and Drop

• take n takes the first n elements of the list.
• For example, take 0 xs = []

• take 3 ”abcde” = ”abc”
• take 3 ”ab” = ”ab”

• Working with infinite list: take 5 [1..] = [1, 2, 3, 4, 5]. Thanks
to normal order (lazy) evaluation.

• Dually, drop n drops the first n elements of the list.
• For example, drop 0 xs = xs
• drop 3 ”abcde” = ”cd”
• drop 3 ”ab” = ””

• take n xs++drop n xs = xs, as long as n ̸= ⊥.

Map and λ

• map :: (a→ b)→ List a→ List b. e.g.
map (1+) [1, 2, 3, 4, 5] = [2, 3, 4, 5, 6].

• map square [1, 2, 3, 4] = [1, 4, 9, 16].
• Every once in a while you may need a small function
which you do not want to give a name to. At such
moments you can use the λ notation:

• map (λx→ x× x) [1, 2, 3, 4] = [1, 4, 9, 16]
• In ASCII λ is written \.

• λ is an important primitive notion. We will talk more
about it later.

Filter

• filter :: (a→ Bool)→ List a→ List a.
• e.g. filter even [2, 7, 4, 3] = [2, 4]
• filter (λn→ n ‘mod‘ 3 == 0) [3, 2, 6, 7] = [3, 6]

• Application: count the number of occurrences of ′a′ in a
list:

• length · filter (’a’ = =)
• Or length · filter (λx→ ’a’ = = x)

• Note a list comprehension can always be translated into
a combination of primitive list generators and map, filter,
and concat.

Zip

• zip :: List a→ List b→ List (a,b)
• e.g. zip ”abcde” [1, 2, 3] = [(’a’, 1), (’b’, 2), (’c’, 3)]
• The length of the resulting list is the length of the shorter
input list.

Positions

• Exercise: define positions :: Char→ String→ List Int, such
that positions x xs returns the positions of occurrences of
x in xs. E.g. positions ’o’ ”roodo” = [1, 2, 4].

• positions x xs = map snd (filter ((x = =) · fst) (zip xs [0..])
• Or,
positions x xs = map snd (filter (λ(y, i)→ x = = y) (zip xs [0..])

• What if you want only the position of the first occurrence
of x?

pos :: Char→ String→ Int
pos x xs = head (positions x xs)

• Due to lazy evaluation (normal order evaluation), positions
of the other occurrences are not evaluated!

Morals of the Story

• Lazy evaluation helps to improve modularity.
• List combinators can be conveniently re-used. Only the
relevant parts are computed.

• The combinator style encourages “wholemeal
programming”.

• Think of aggregate data as a whole, and process them as a
whole!

λ expressions

• λx→ e denotes a function whose argument is x and
whose body is e.

• (λx→ e1) e2 denotes the function (λx→ e1) applied to e2.
It can be reduced to e1 with its free occurrences of x
replaced by e2.

• E.g.

(λx→ x× x) (3+ 4)
= (3+ 4)× (3+ 4)
= 49 .

• λ expression is a primitive and essential notion. Many
other constructs can be seen as syntax sugar of λ’s.

• For example, our previous definition of square can be
seen as an abbreviation of

square :: Int→ Int
square = λx→ x× x .

• Indeed, square is merely a value that happens to be a
function, which is in turn given by a λ expression.

• λ’s are like all values — they can appear inside an
expression, be passed as parameters, returned as results,
etc.

• In fact, it is possible to build a complete programming
language consisting of only λ expressions and
applications. Look up “λ calculus”.

• λx→ λy→ e is abbreviated to λx y→ e.
• The following definitions are all equivalent:

smaller x y = if x ≤ y then x else y
smaller x = λy→ if x ≤ y then x else y
smaller = λx→ λy→ if x ≤ y then x else y
smaller = λx y→ if x ≤ y then x else y .

Replacing Constructors

• The function foldr is among the most important functions
on lists.

foldr :: (a→ b→ b)→ b→ List a→ b

• One way to look at foldr (⊕) e is that it replaces [ ] with e
and (:) with (⊕):

foldr (⊕) e [1, 2, 3, 4]
= foldr (⊕) e (1 : (2 : (3 : (4 : [ ]))))

= 1⊕ (2⊕ (3⊕ (4⊕ e))).
• sum = foldr (+) 0.
• One can see that id = foldr (:) [ ].

Some Trivial Folds on Lists

• Function maximum returns the maximum element in a list:

• maximum = foldr max -∞.

• Function prod returns the product of a list:

• prod = foldr (×) 1.

• Function and returns the conjunction of a list:

• and = foldr (∧) True.

• Lets emphasise again that id on lists is a fold:

• id = foldr (:) [ ].

Some Slightly Complex Folds

• length = foldr (λx n→ 1+ n) 0.
• map f = foldr (λx xs→ f x : xs) [ ].
• xs++ ys = foldr (:) ys xs. Compare this with id!
• filter p = foldr (fil p) [ ]

where fil p x xs = if p x then (x : xs) else xs.

The Ubiquitous Fold

• In fact, any function that takes a list as its input can be
written in terms of foldr — although it might not be always
practical.

• With fold it comes one of the most important theorem in
program calculation — the fold-fusion theorem. We will
talk about it later.

Fold-Left

There is another, sometimes useful fold on lists: foldl.

foldl :: (b→ a→ b)→ b→ List a→ b .

One may see from its type that it brackets the elements of the
given list from the different direction:

foldl (⊕) e [1, 2, 3, 4]
= (((e⊕ 1)⊕ 2)⊕ 3)⊕ 4.

It has advantages for some applications. We will talk about it
in the last lecture.

Definition and Proof by Induction



Total Functional Programming

• The next few lectures concerns inductive definitions and
proofs of datatypes and programs.

• While Haskell provides allows one to define
nonterminating functions, infinite data structures, for now
we will only consider its total, finite fragment.

• That is, we temporarily
• consider only finite data structures,
• demand that functions terminate for all value in its input
type, and

• provide guidelines to construct such functions.

• Infinite datatypes and non-termination will be discussed
later in this course.

The So-Called “Mathematical Induction”

• Let P be a predicate on natural numbers.
• We’ve all learnt this principle of proof by induction: to
prove that P holds for all natural numbers, it is sufficient
to show that

• P 0 holds;
• P (1+ n) holds provided that Pn does.

Proof by Induction on Natural Numbers

• We can see the above inductive principle as a result of
seeing natural numbers as defined by the datatype 1

data Nat = 0 | 1+ Nat .

• That is, any natural number is either 0, or 1+ n where n is
a natural number.

• In this lecture, 1+ is written in bold font to emphasise that
it is a data constructor (as opposed to the function (+), to
be defined later, applied to a number 1).

1Not a real Haskell definition.

A Proof Generator

Given P 0 and Pn⇒ P (1+ n), how does one prove, for example,
P 3?

P (1+ (1+ (1+ 0)))
⇐ { P (1+ n)⇐ Pn }

P (1+ (1+ 0))
⇐ { P (1+ n)⇐ Pn }

P (1+ 0)
⇐ { P (1+ n)⇐ Pn }

P 0 .

Having done math. induction can be seen as having designed
a program that generates a proof — given any n :: Nat we can
generate a proof of Pn in the manner above.

Inductively Defined Functions

• Since the type Nat is defined by two cases, it is natural to
define functions on Nat following the structure:

exp :: Nat→ Nat→ Nat
exp b 0 = 1
exp b (1+ n) = b× exp b n .

• Even addition can be defined inductively

(+) :: Nat→ Nat→ Nat
0+ n = n
(1+ m) + n = 1+ (m+ n) .

• Exercise: define (×)?

A Value Generator

Given the definition of exp, how does one compute exp b 3?

exp b (1+ (1+ (1+ 0)))
= { definition of exp }

b× exp b (1+ (1+ 0))
= { definition of exp }

b× b× exp b (1+ 0)
= { definition of exp }

b× b× b× exp b 0
= { definition of exp }

b× b× b× 1 .

It is a program that generates a value, for any n :: Nat.
Compare with the proof of P above.

Moral: Proving is Programming

An inductive proof is a program that generates a proof for any
given natural number.

An inductive program follows the same structure of an
inductive proof.

Proving and programming are very similar activities.

Without the n+ k Pattern

• Unfortunately, newer versions of Haskell abandoned the
“n+ k pattern” used in the previous slides:

exp :: Int→ Int→ Int
exp b 0 = 1
exp b n = b× exp b (n− 1) .

• Nat is defined to be Int in MiniPrelude.hs. Without
MiniPrelude.hs you should use Int.

• For the purpose of this course, the pattern 1+ n reveals
the correspondence between Nat and lists, and matches
our proof style. Thus we will use it in the lecture.

• Remember to remove them in your code.

49 / 214



Proof by Induction

• To prove properties about Nat, we follow the structure as
well.

• E.g. to prove that exp b (m+ n) = exp b m× exp b n.
• One possibility is to preform induction on m. That is,
prove Pm for all m :: Nat, where
Pm ≡ (∀n :: exp b (m+ n) = exp b m× exp b n).

Proof by Induction

Recall Pm ≡ (∀n :: exp b (m+ n) = exp b m× exp b n).

Case m := 0. For all n, we reason:
exp b (0+ n)

= { defn. of (+) }
exp b n

= { defn. of (×) }
1× exp b n

= { defn. of exp }
exp b 0× exp b n .

We have thus proved P 0.

Proof by Induction

Recall Pm ≡ (∀n :: exp b (m+ n) = exp b m× exp b n).

Case m := 1+ m. For all n, we reason:

exp b ((1+ m) + n)

= { defn. of (+) }
exp b (1+ (m+ n))

= { defn. of exp }
b× exp b (m+ n)

= { induction }
b× (exp b m× exp b n)

= { (×) associative }
(b× exp b m)× exp b n

= { defn. of exp }
exp b (1+ m)× exp b n .

We have thus proved P (1+ m), given P m.

Structure Proofs by Programs

• The inductive proof could be carried out smoothly,
because both (+) and exp are defined inductively on its
lefthand argument (of type Nat).

• The structure of the proof follows the structure of the
program, which in turns follows the structure of the
datatype the program is defined on.

Lists and Natural Numbers

• We have yet to prove that (×) is associative.
• The proof is quite similar to the proof for associativity of
(++), which we will talk about later.

• In fact, Nat and lists are closely related in structure.
• Most of us are used to think of numbers as atomic and
lists as structured data. Neither is necessarily true.

• For the rest of the course we will demonstrate induction
using lists, while taking the properties for Nat as given.

An Inductively Defined Set?

• For a set to be “inductively defined”, we usually mean that
it is the smallest fixed-point of some function.

• What does that maen?

Fixed-Point and Prefixed-Point

• A fixed-point of a function f is a value x such that f x = x.
• Theorem. f has fixed-point(s) if f is a monotonic function
defined on a complete lattice.

• In general, given f there may be more than one fixed-point.
• A prefixed-point of f is a value x such that f x ≤ x.

• Apparently, all fixed-points are also prefixed-points.

• Theorem. the smallest prefixed-point is also the smallest
fixed-point.

Example: Nat

• Recall the usual definition: Nat is defined by the following
rules:
1. 0 is in Nat;
2. if n is in Nat, so is 1+ n;
3. there is no other Nat.

• If we define a function F from sets to sets:
F X = {0} ∪ {1+ n | n ∈ X}, 1. and 2. above means that
FNat ⊆ Nat. That is, Nat is a prefixed-point of F.

• 3. means that we want the smallest such prefixed-point.
• Thus Nat is also the least (smallest) fixed-point of F.

Least Prefixed-Point

Formally, let F X = {0} ∪ {1+ n | n ∈ X}, Nat is a set such that

FNat ⊆ Nat , (1)
(∀X : F X ⊆ X ⇒ Nat ⊆ X) , (2)

where (1) says that Nat is a prefixed-point of F, and (2) it is the
least among all prefixed-points of F.

Mathematical Induction, Formally

• Given property P, we also denote by P the set of elements
that satisfy P.

• That P 0 and Pn⇒ P (1+n) is equivalent to {0} ⊆ P and
{1+ n | n ∈ P} ⊆ P,

• which is equivalent to F P ⊆ P. That is, P is a
prefixed-point!

• By (2) we have Nat ⊆ P. That is, all Nat satisfy P!
• This is “why mathematical induction is correct.”

59 / 214



Coinduction?

There is a dual technique called coinduction where, instead of
least prefixed-points, we talk about greatest postfixed points.
That is, largest x such that x ≤ f x.

With such construction we can talk about infinite data
structures.

Inductively Defined Lists

• Recall that a (finite) list can be seen as a datatype defined
by: 2

data List a = [] | a : List a .

• Every list is built from the base case [ ], with elements
added by (:) one by one: [1, 2, 3] = 1 : (2 : (3 : [ ])).

2Not a real Haskell definition.

All Lists Today are Finite

But what about infinite lists?

• For now let’s consider finite lists only, as having infinite
lists make the semantics much more complicated. 3

• In fact, all functions we talk about today are total
functions. No ⊥ involved.

3What does that mean? We will talk about it later.

Set-Theoretically Speaking...

The type List a is the smallest set such that

1. [ ] is in List a;
2. if xs is in List a and x is in a, x : xs is in List a as well.

Inductively Defined Functions on Lists

• Many functions on lists can be defined according to how a
list is defined:

sum :: List Int→ Int
sum [ ] = 0
sum (x : xs) = x+ sum xs .

map :: (a→ b)→ List a→ List b
map f [ ] = [ ]

map f (x : xs) = F X : map f xs .

List Append

• The function (++) appends two lists into one

(++) :: List a→ List a→ List a
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs++ ys) .

• Compare the definition with that of (+)!

Proof by Structural Induction on Lists

• Recall that every finite list is built from the base case [ ],
with elements added by (:) one by one.

• To prove that some property P holds for all finite lists, we
show that
1. P [ ] holds;
2. forall x and xs, P (x : xs) holds provided that P xs holds.

For a Particular List...

Given P [ ] and P xs⇒ P (x : xs), for all x and xs, how does one
prove, for example, P [1, 2, 3]?

P (1 : 2 : 3 : [ ])
⇐ { P (x : xs)⇐ P xs }

P (2 : 3 : [ ])
⇐ { P (x : xs)⇐ P xs }

P (3 : [ ])
⇐ { P (x : xs)⇐ P xs }

P [ ] .

Appending is Associative

To prove that xs++(ys++ zs) = (xs++ ys)++ zs.

Let P xs = (∀ys, zs :: xs++(ys++ zs) = (xs++ ys)++ zs), we prove
P by induction on xs.

Case xs := [ ]. For all ys and zs, we reason:

[ ] ++(ys++ zs)
= { defn. of (++) }

ys++ zs
= { defn. of (++) }

([ ] ++ ys)++ zs .

We have thus proved P [].

Appending is Associative

Case xs := x : xs. For all ys and zs, we reason:

(x : xs)++(ys++ zs)
= { defn. of (++) }

x : (xs++(ys++ zs))
= { induction }

x : ((xs++ ys)++ zs)
= { defn. of (++) }

(x : (xs++ ys))++ zs
= { defn. of (++) }

((x : xs)++ ys)++ zs .

We have thus proved P (x : xs), given P xs.

Do We Have To Be So Formal?

• In our style of proof, every step is given a reason. Do we
need to be so pedantic?

• Being formal helps you to do the proof:
• In the proof of exp b (m+ n) = exp b m× exp b n, we
expect that we will use induction to somewhere. Therefore
the first part of the proof is to generate exp b (m+ n).

• In the proof of associativity, we were working toward
generating xs++(ys++ zs).

• By being formal we can work on the form, not the
meaning. Like how we solved the knight/knave problem

• Being formal actually makes the proof easier!
• Make the symbols do the work.

Length

• The function length defined inductively:

length :: List a→ Nat
length [ ] = 0
length (x : xs) = 1+ (length xs) .

• Exercise: prove that length distributes into (++):

length (xs++ ys) = length xs+ length ys

Concatenation

• While (++) repeatedly applies (:), the function concat
repeatedly calls (++):

concat :: List (List a)→ List a
concat [ ] = [ ]

concat (xs : xss) = xs++ concat xss .
• Compare with sum.
• Exercise: prove sum · concat = sum ·map sum.

Definition by Induction/Recursion

• Rather than giving commands, in functional programming
we specify values; instead of performing repeated actions,
we define values on inductively defined structures.

• Thus induction (or in general, recursion) is the only
“control structure” we have. (We do identify and abstract
over plenty of patterns of recursion, though.)

• To inductively define a function f on lists, we specify a
value for the base case (f [ ]) and, assuming that f xs has
been computed, consider how to construct f (x : xs) out of
f xs.

Filter

• filter p xs keeps only those elements in xs that satisfy p.

filter :: (a→ Bool)→ List a→ List a
filter p [ ] = [ ]

filter p (x : xs) | p x = x : filter p xs
| otherwise = filter p xs .

Take and Drop

• Recall take and drop, which we used in the previous
exercise.

take :: Nat→ List a→ List a
take 0 xs = [ ]

take (1+ n) [ ] = [ ]

take (1+ n) (x : xs) = x : take n xs .

drop :: Nat→ List a→ List a
drop 0 xs = xs
drop (1+ n) [ ] = [ ]

drop (1+ n) (x : xs) = drop n xs .
• Prove: take n xs++drop n xs = xs, for all n and xs.

TakeWhile and DropWhile

• takeWhile p xs yields the longest prefix of xs such that p
holds for each element.

takeWhile :: (a→ Bool)→ List a→ List a
takeWhile p [ ] = [ ]

takeWhile p (x : xs) | p x = x : takeWhile p xs
| otherwise = [ ] .

• dropWhile p xs drops the prefix from xs.

dropWhile :: (a→ Bool)→ List a→ List a
dropWhile p [ ] = [ ]

dropWhile p (x : xs) | p x = dropWhile p xs
| otherwise = x : xs .

• Prove: takeWhile p xs++dropWhile p xs = xs.

List Reversal

• reverse [1, 2, 3, 4] = [4, 3, 2, 1].

reverse :: List a→ List a
reverse [ ] = [ ]

reverse (x : xs) = reverse xs++[x] .

All Prefixes and Suffixes

• inits [1, 2, 3] = [[ ], [1], [1, 2], [1, 2, 3]]

inits :: List a→ List (List a)
inits [ ] = [[ ]]

inits (x : xs) = [ ] : map (x :) (inits xs) .
• tails [1, 2, 3] = [[1, 2, 3], [2, 3], [3], [ ]]

tails :: List a→ List (List a)
tails [ ] = [[ ]]

tails (x : xs) = (x : xs) : tails xs .

Totality

• Structure of our definitions so far:
f [ ] = . . .

f (x : xs) = . . . f xs . . .
• Both the empty and the non-empty cases are covered,
guaranteeing there is a matching clause for all inputs.

• The recursive call is made on a “smaller” argument,
guranteeing termination.

• Together they guarantee that every input is mapped to
some output. Thus they define total functions on lists.

Variations with the Base Case

• Some functions discriminate between several base cases.
E.g.

fib :: Nat→ Nat
fib 0 = 0
fib 1 = 1
fib (2+ n) = fib (1+n) + fib n .

80 / 214



• Some functions make more sense when it is defined only
on non-empty lists:

f [x] = . . .

f (x : xs) = . . .

• What about totality?
• They are in fact functions defined on a different datatype:

data List+ a = Singleton a | a : List+ a .

• We do not want to define map, filter again for List+ a. Thus
we reuse List a and pretend that we were talking about
List+ a.

• It’s the same with Nat. We embedded Nat into Int.
• Ideally we’d like to have some form of subtyping. But that
makes the type system more complex.

81 / 214



Lexicographic Induction

• It also occurs often that we perform lexicographic
induction on multiple arguments: some arguments
decrease in size, while others stay the same.

• E.g. the function merge merges two sorted lists into one
sorted list:

merge :: List Int→ List Int→ List Int
merge [ ] [ ] = [ ]

merge [ ] (y : ys) = y : ys
merge (x : xs) [ ] = x : xs
merge (x : xs) (y : ys)
| x ≤ y = x : merge xs (y : ys)
| otherwise = y : merge (x : xs) ys .

Zip

Another example:

zip :: List a→ List b→ List (a,b)
zip [ ] [ ] = [ ]

zip [ ] (y : ys) = [ ]

zip (x : xs) [ ] = [ ]

zip (x : xs) (y : ys) = (x, y) : zip xs ys .

Non-Structural Induction

• In most of the programs we’ve seen so far, the recursive
call are made on direct sub-components of the input (e.g.
f (x : xs) = ..f xs..). This is called structural induction.

• It is relatively easy for compilers to recognise structural
induction and determine that a program terminates.

• In fact, we can be sure that a program terminates if the
arguments get “smaller” under some (well-founded)
ordering.

Mergesort

• In the implemenation of mergesort below, for example,
the arguments always get smaller in size.

msort :: List Int→ List Int
msort [ ] = [ ]

msort [x] = [x]
msort xs = merge (msort ys) (msort zs) ,
where n = length xs ‘div‘ 2

ys = take n xs
zs = drop n xs .

• What if we omit the case for [x]?

• If all cases are covered, and all recursive calls are applied
to smaller arguments, the program defines a total
function.

A Non-Terminating Definition

• Example of a function, where the argument to the
recursive does not reduce in size:

f :: Int→ Int
f 0 = 0
f n = f n .

• Certainly f is not a total function. Do such definitions
“mean” something? We will talk about these later.

Internally Labelled Binary Trees

• This is a possible definition of internally labelled binary
trees:

data Tree a = Null | Node a (Tree a) (Tree a) ,

• on which we may inductively define functions:

sumT :: Tree Nat→ Nat
sumT Null = 0
sumT (Node x t u) = x+ sumT t+ sumT u .

Exercise: given (↓) :: Nat→ Nat→ Nat, which yields the
smaller one of its arguments, define the following functions

1. minT :: Tree Nat→ Nat, which computes the minimal
element in a tree.

2. mapT :: (a→ b)→ Tree a→ Tree b, which applies the
functional argument to each element in a tree.

3. Can you define (↓) inductively on Nat? 4

4In the standard Haskell library, (↓) is called min.

Induction Principle for Tree

• What is the induction principle for Tree?
• To prove that a predicate P on Tree holds for every tree, it
is sufficient to show that

1. P Null holds, and;
2. for every x, t, and u, if P t and P u holds, P (Node x t u)
holds.

• Exercise: prove that for all n and t,
minT (mapT (n+) t) = n+minT t. That is,
minT ·mapT (n+) = (n+) ·minT.

Induction Principle for Other Types

• Recall that data Bool = False | True. Do we have an
induction principle for Bool?

• To prove a predicate P on Bool holds for all booleans, it is
sufficient to show that

1. P False holds, and
2. P True holds.

• Well, of course.

• What about (A× B)? How to prove that a predicate P on
(A× B) is always true?

• One may prove some property P1 on A and some property
P2 on B, which together imply P.

• That does not say much. But the “induction principle” for
products allows us to extract, from a proof of P, the proofs
P1 and P2.

• Every inductively defined datatype comes with its
induction principle.

• We will come back to this point later.

Program Calculation



A Quick Review

• Functions are the basic building blocks. They may be
passed as arguments, may return functions, and can be
composed together.

• While one issues commands in an imperative language, in
functional programming we specify values, and computers
try to reduce the values to their normal forms.

• Formal reasoning: reasoning with the form (syntax) rather
than the semantics. Let the symbols do the work!

• ‘Wholemeal’ programming: think of aggregate data as a
whole, and process them as a whole.

A Quick Review

• Once you describe the values as algebraic datatypes, most
programs write themselves through structural recursion.

• Programs and their proofs are closely related. They share
similar structure, by induction over input data.

• Properties of programs can be reasoned about in
equations, just like high school algebra.

Data Representation

• So far we have (surprisingly) been talking about
mathematics without much concern regarding efficiency.
Time for a change.

• Take lists for example. Recall the definition:
data List a = [ ] | a : List a.

• Our representation of lists is biased. The left most
element can be fetched immediately.

• Thus. (:), head, and tail are constant-time operations,
while init and last takes linear-time.

• In most implementations, the list is represented as a
linked-list.

List Concatenation Takes Linear Time

• Recall (++):

[ ] ++ ys =

ys

(x : xs)++ ys =

x : (xs++ ys)

• Consider [1, 2, 3] ++[4, 5]:

(1 : 2 : 3 : [ ])++(4 : 5 : [ ])
= 1 : ((2 : 3 : [ ])++(4 : 5 : [ ]))
= 1 : 2 : ((3 : [ ])++(4 : 5 : [ ]))
= 1 : 2 : 3 : ([ ] ++(4 : 5 : [ ]))
= 1 : 2 : 3 : 4 : 5 : [ ]

• (++) runs in time proportional to the length of its left
argument.

Full Persistency

• Compound data structures, like simple values, are just
values, and thus must be fully persistent.

• That is, in the following code:

let xs = [1, 2, 3]
ys = [4, 5]
zs = xs++ ys

in . . .body . . .
• The body may have access to all three values. Thus ++
cannot perform a destructive update.

Linked v.s. Block Data Structures

• Trees are usually represented in a similar manner, through
links.

• Fully persistency is easier to achieve for such linked data
structures.

• Accessing arbitrary elements, however, usually takes linear
time.

• In imperative languages, constant-time random access is
usually achieved by allocating lists (usually called arrays
in this case) in a consecutive block of memory.

Linked v.s. Block Data Structures

• Consider the following code, where xs is an array
(implemented as a block), and ys is like xs, apart from its
10th element:

let xs = [1..100]
ys = update xs 10 20

in . . .body . . .
• To allow access to both xs and ys in body, the update
operation has to duplicate the entire array.

• Thus people have invented some smart data structure to
do so, in around O(log n) time.

• On the other hand, update may simply overwrite xs if we
can somehow make sure that nobody other than ys uses
xs.

• Both are advanced topics, however.

Another Linear-Time Operation

• Taking all but the last element of a list:

init [x] =

[ ]

init (x : xs) =

x : init xs

• Consider init [1, 2, 3, 4]:

init (1 : 2 : 3 : 4 : [ ])

= 1 : init (2 : 3 : 4 : [ ])

= 1 : 2 : init (3 : 4 : [ ])

= 1 : 2 : 3 : init (4 : [ ])

= 1 : 2 : 3 : [ ]

Sum, Map, etc

• Functions like sum, maximum, etc. needs to traverse
through the list once to produce a result. So their running
time is definitely O(n).

• If f takes time O(t), map f takes time O(n× t) to complete.
Similarly with filter p.

• In a lazy setting, map f produces its first result in O(t) time.
We won’t need lazy features for now, however.

Sum of Squares

• Given a sequence a1,a2,…,an, compute a21 + a22 + . . .+ a2n.
Specification: sumsq = sum ·map square.

• The spec. builds an intermediate list. Can we eliminate it?
• The input is either empty or not. When it is empty:

sumsq [ ]

= { definition of sumsq }
(sum ·map square) [ ]

= { function composition }
sum (map square [ ])

= { definition of map }
sum [ ]

= { definition of sum }
0

Alternative Definition for sumsq

• From sumsq = sum ·map square, we have proved that

sumsq [ ] = 0
sumsq (x : xs) = square x+ sumsq xs

• Equivalently, we have shown that sum ·map square is a
solution of

f [ ] = 0
f (x : xs) = square x+ f xs

• However, the solution of the equations above is unique.
• Thus we can take it as another definition of sumsq.
Denotationally it is the same function; operationally, it is
(slightly) quicker.

• Exercise: try calculating an inductive definition of count.

How Far Can We Get?

• Specification of maximum segment sum:

mss :: List Int→ Int
mss = maximum ·map sum · segments
segments :: List a→ List (List a)
segments = concat ·map inits · tails

• From the specification we can calculate a linear time
algorithm.

Remark: Why Functional Programming?

• Time to muse on the merits of functional programming.
Why functional programming?

• Algebraic datatype? List comprehension? Lazy evaluation?
Garbage collection? These are just language features that
can be migrated.

• No side effects.5 But why taking away a language feature?
• By being pure, we have a simpler semantics in which we
are allowed to construct and reason about programs.

• In an imperative language we do not even have
f 4+ f 4 = 2× f 4.

• Ease of reasoning. That’s the main benefit we get.

5Unless introduced in a disciplined way.

Steep Lists

• A steep list is a list in which every element is larger than
the sum of those to its right:

steep :: List Int→ Bool
steep [ ] = True
steep (x : xs) = steep xs ∧ x > sum xs.

• The definition above, if executed directly, is an O(n2)
program. Can we do better?

• Just now we learned to construct a generalised function
which takes more input. This time, we try the dual
technique: to construct a function returning more results.

Generalise by Returning More

• Recall that fst (a,b) = a and snd (a,b) = b.
• It is hard to quickly compute steep alone. But if we define

steepsum xs = (steep xs, sum xs),

• and manage to synthesise a quick definition of steepsum,
we can implement steep by steep = fst · steepsum.

• We again proceed by case analysis. Trivially,

steepsum [ ] = (True, 0).

Deriving for the Non-Empty Case

For the case for non-empty inputs:

steepsum (x : xs)

= { definition of steepsum }
(steep (x : xs), sum (x : xs))

= { definitions of steep and sum }
(steep xs ∧ x > sum xs, x+ sum xs)

= { extracting sub-expressions involving xs }
let (b, y) = (steep xs, sum xs)
in (b ∧ x > y, x+ y)

= { definition of steepsum }
let (b, y) = steepsum xs
in (b ∧ x > y, x+ y).

Synthesised Program

We have thus come up with a O(n) time program:

steep = fst · steepsum
steepsum [ ] = (True, 0)
steepsum (x : xs) = let (b, y) = steepsum xs

in (b ∧ x > y, x+ y),

Being Quicker by Doing More?

• A more generalised program can be implemented more
efficiently?

• A common phenomena! Sometimes the less general
function cannot be implemented inductively at all!

• It also often happens that a theorem needs to be
generalised to be proved. We will see that later.

• An obvious question: how do we know what generalisation
to pick?

• There is no easy answer — finding the right generalisation
one of the most difficulty act in programming!

• Sometimes we simply generalise by examining the form of
the formula.

Reversing a List

• The function reverse is defined by:

reverse [ ] = [ ],
reverse (x : xs) = reverse xs++[x].

• E.g.
reverse [1, 2, 3, 4] = ((([ ] ++[4])++[3])++[2])++[1] = [4, 3, 2, 1].

• But how about its time complexity? Since (++) is O(n), it
takes O(n2) time to revert a list this way.

• Can we make it faster?

Introducing an Accumulating Parameter

• Let us consider a generalisation of reverse. Define:

revcat :: [a]→ [a]→ [a]
revcat xs ys = reverse xs++ ys.

• If we can construct a fast implementation of revcat, we
can implement reverse by:

reverse xs = revcat xs [ ].

Reversing a List, Base Case

Let us use our old trick. Consider the case when xs is [ ]:

revcat [ ] ys

= { definition of revcat }
reverse [ ] ++ ys

= { definition of reverse }
[ ] ++ ys

= { definition of (++) }
ys.

Reversing a List, Inductive Case

Linear-Time List Reversal

• We have therefore constructed an implementation of
revcat which runs in linear time!

revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys).

• A generalisation of reverse is easier to implement than
reverse itself? How come?

• If you try to understand revcat operationally, it is not
difficult to see how it works.

• The partially reverted list is accumulated in ys.
• The initial value of ys is set by reverse xs = revcat xs [ ].
• Hmm... it is like a loop, isn’t it?

Tracing Reverse

reverse [1, 2, 3, 4]
= revcat [1, 2, 3, 4] [ ]
= revcat [2, 3, 4] [1]
= revcat [3, 4] [2, 1]
= revcat [4] [3, 2, 1]
= revcat [ ] [4, 3, 2, 1]
= [4, 3, 2, 1]

reverse xs = revcat xs [ ]
revcat [ ] ys = ys
revcat (x : xs) ys = revcat xs (x : ys)

xs, ys ← XS, [ ];
while xs ̸= [ ] do

xs, ys ← (tail xs), (head xs : ys);
return ys

Tail Recursion

• Tail recursion: a special case of recursion in which the last
operation is the recursive call.

f x1 . . . xn = {base case}
f x1 . . . xn = f x′1 . . . x′n

• To implement general recursion, we need to keep a stack
of return addresses. For tail recursion, we do not need
such a stack.

• Tail recursive definitions are like loops. Each xi is updated
to x′i in the next iteration of the loop.

• The first call to f sets up the initial values of each xi.

Accumulating Parameters

• To efficiently perform a computation (e.g. reverse xs), we
introduce a generalisation with an extra parameter, e.g.:

revcat xs ys = reverse xs++ ys.

• Try to derive an efficient implementation of the
generalised function. The extra parameter is usually used
to “accumulate” some results, hence the name.

• To make the accumulation work, we usually need some
kind of associativity.

• A technique useful for, but not limited to, constructing
tail-recursive definition of functions.

Accumulating Parameter: Another Example

• Recall the “sum of squares” problem:
sumsq [ ] = 0
sumsq (x : xs) = square x+ sumsq xs.

• The program still takes linear space (for the stack of
return addresses). Let us construct a tail recursive
auxiliary function.

• Introduce ssp xs n =

sumsq xs+ n

.
• Initialisation: sumsq xs =

ssp xs 0

.
• Construct ssp:

ssp [ ] n = 0+ n = n
ssp (x : xs) n = (square x+ sumsq xs) + n

= sumsq xs+ (square x+ n)
= ssp xs (square x+ n).

Labelling a List

• Consider the task of labelling elements in a list with its
index.

index :: List a→ List (Int,a)
index = zip [0..]

• To construct an inductive definition, the case for [ ] is easy.
For the x : xs case:

index (x : xs)
= zip [0..] (x : xs)
= (0, x) : zip [1..] xs

• Alas, zip [1..] cannot be fold back to index!
• What if we turn the varying part into…a variable?

Labelling a List, Second Attempt

• Introduce idxFrom :: List a→ Int→ List (Int,a):

idxFrom xs n = zip [n..] xs

• Initialisation: index xs =

idxFrom xs 0

.

• We reason:
idxFrom (x : xs) n

= zip [n..] (x : xs)
= (n, x) : zip [n+ 1..] xs
= (n, x) : idxFrom xs (n+ 1)

Summing Up a List in Reverse

• Prove: sum · reverse = sum, using the definition
reverse xs = revcat xs [ ]. That is, proving
sum (revcat xs [ ]) = sum xs.

• Base case trivial. For the case x : xs:
sum (reverse (x : xs))

= sum (revcat (x : xs) [ ])
= sum (revcat xs [x])

• Then we are stuck, since we cannot use the induction
hypothesis sum (revcat xs [ ]) = sum xs.

• Again, generalise [x] to a variable.

Summing Up a List in Reverse, Second Attempt

• Second attempt: prove a lemma:

sum (revcat xs ys) =

sum xs+ sum ys

• By letting ys = [ ] we get the previous property.

• For the case x : xs we reason:
sum (revcat (x : xs) ys)

= sum (revcat xs (x : ys))
= { induction hypothesis }

sum xs+ sum (x : ys)
= sum xs+ x+ sum ys
= sum (x : xs) + sum ys

Work Less by Proving More

• A stronger theorem is easier to prove! Why is that?
• By strengthening the theorem, we also have a stronger
induction hypothesis, which makes an inductive proof
possible.

• Finding the right generalisation is an art — it’s got to be
strong enough to help the proof, yet not too strong to be
provable.

• The same with programming. By generalising a function
with additional arguments, it is passed more information
it may use, thus making an inductive definition possible.

• The speeding up of revcat, in retrospect, is an accidental
“side effect” — revcat, being inductive, goes through the list
only once, and is therefore quicker.

A Real Case

• A property I actually had to prove for a paper:

smsp (trim (x : xs)) = smsp (trim (x : win xs))
⇐ smsp (trim (x : xs)) >d mds xs

• It took me a week to construct the right generalisation:

smsp (trim (zs++ xs)) = smsp (trim (zs++win xs))
⇐ smsp (trim (zs++ xs)) >d mds xs

• Once the right property is found, the actual proof was
done in about 20 minutes.

• “Someone once described research as ‘finding out
something to find out, then finding it out’; the first part is
often harder than the second.”

Remark

• The sum · reverse example is superficial — the same
property is much easier to prove using the O(n2)-time
definition of reverse.

• That’s one of the reason we defer the discussion about
efficiency — to prove properties of a function we
sometimes prefer to roll back to a slower version.

• In our exercises there is an example where you need
revcat to prove a property about reverse.

• Show that reverse · reverse = id

A Common Pattern We’ve Seen Many Times…

sum [ ] = 0
sum (x : xs) = x+ sum xs

length [ ] = 0
length (x : xs) = 1+ length xs

map f [ ] = [ ]

map f (x : xs) = f x : map f xs

This pattern is extracted and called foldr:

foldr f e [ ] = e,
foldr f e (x : xs) = f x (foldr f e xs).

Replacing Constructors

foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs)

One way to look at foldr (⊕) e is that it replaces [ ] with e and
(:) with (⊕):

foldr (⊕) e [1, 2, 3, 4]
= foldr (⊕) e (1 : (2 : (3 : (4 : [ ]))))

= 1⊕ (2⊕ (3⊕ (4⊕ e))).

• sum = foldr (+) 0.
• length = foldr (λx n.1+ n) 0.
• map f = foldr (λx xs.f x : xs) [ ].
• One can see that id = foldr (:) [ ].

Some Trivial Folds on Lists

Function max returns the maximum element in a list:
•

maximum [ ] = -∞,
maximum (x : xs) = x ↑ maximum xs.

• maximum = foldr (↑) -∞.

Function prod returns the product of a list:
•

product [ ] = 1,
product (x : xs) = x× product xs.

• product = foldr (×) 1.

Function and returns the conjunction of a list:
•

and [ ] = true,
and (x : xs) = x ∧ and xs.

• and = foldr (∧) true.

Lets emphasise again that id on lists is a fold:
•

id [ ] = [ ],
id (x : xs) = x : id xs.

• id = foldr (:) [ ].

•

and [ ] = true,
and (x : xs) = x ∧ and xs.

• and = foldr (∧) true.

Lets emphasise again that id on lists is a fold:
•

id [ ] = [ ],
id (x : xs) = x : id xs.

• id = foldr (:) [ ].

Some Functions We Have Seen…

(++ ys) = foldr (:) ys

.

(++) :: [a]→ [a]→ [a]
[ ] ++ ys = ys
(x : xs)++ ys = x : (xs++ ys) .

concat =

foldr (++) [ ]

.
concat :: [[a]]→ [a]
concat [ ] = [ ]

concat (xs : xss) = xs++ concat xss .

Replacing Constructors

• Understanding foldr from its type. Recall

data [a] = [ ] | a : [a] .

• Types of the two constructors: [ ] :: [a], and
(:) :: a→ [a]→ [a].

• foldr replaces the constructors:

foldr :: (a→ b→ b)→ b→ [a]→ b
foldr f e [ ] = e
foldr f e (x : xs) = f x (foldr f e xs) .

Why Folds?

• “What are the three most important factors in a
programming language?”

Abstraction, abstraction, and
abstraction!

• Control abstraction, procedure abstraction, data
abstraction,…can programming patterns be abstracted too?

• Program structure becomes an entity we can talk about,
reason about, and reuse.

• We can describe algorithms in terms of fold, unfold, and
other recognised patterns.

• We can prove properties about folds,
• and apply the proved theorems to all programs that are
folds, either for compiler optimisation, or for mathematical
reasoning.

• Among the theorems about folds, the most important is
probably the fold-fusion theorem.

The Fold-Fusion Theorem

The theorem is about when the composition of a function and
a fold can be expressed as a fold.

Theorem (foldr-Fusion)
Given f :: a→ b→ b, e :: b, h :: b→ c, and g :: a→ c→ c, we
have:

h · foldr f e = foldr g (h e) ,

if h (f x y) = g x (h y) for all x and y.

For program derivation, we are usually given h, f, and e, from
which we have to construct g.

Tracing an Example

Let us try to get an intuitive understand of the theorem:

h (foldr f e [a,b, c])
= { definition of foldr }

h (f a (f b (f c e)))

= { since h (f x y) = g x (h y) }
g a (h (f b (f c e)))

= { since h (f x y) = g x (h y) }
g a (g b (h (f c e)))

= { since h (f x y) = g x (h y) }
g a (g b (g c (h e)))

= { definition of foldr }
foldr g (h e) [a,b, c] .

Sum of Squares, Again

• Consider sum ·map square again. This time we use the
fact that map f = foldr (mf f) [ ], where mf f x xs = f x : xs.

• sum ·map square is a fold, if we can find a ssq such that
sum (mf square x xs) = ssq x (sum xs). Let us try:

sum (mf square x xs)
= { definition of mf }

sum (square x : xs)
= { definition of sum }

square x+ sum xs
= { let ssq x y = square x+ y }

ssq x (sum xs) .

Therefore, sum ·map square = foldr ssq 0.

Sum of Squares, without Folds

Recall that this is how we derived the inductive case of sumsq
yesterday:

sumsq (x : xs)
= { definition of sumsq }

sum (map square (x : xs))
= { definition of map }

sum (square x : map square xs)
= { definition of sum }

square x+ sum (map square xs)
= { definition of sumsq }

square x+ sumsq xs .

Comparing the two derivations, by using fold-fusion we supply
only the “important” part. 140 / 214



More on Folds and Fold-fusion

• Compare the proof with the one yesterday. They are
essentially the same proof.

• Fold-fusion theorem abstracts away the common parts in
this kind of inductive proofs, so that we need to supply
only the “important” parts.

• Tupling can be seen as a kind of fold-fusion. The
derivation of steepsum, for example, can be seen as
fusing:

steepsum · id = steepsum · foldr (:) [ ].
• Recall that steepsum xs = (steep xs, sum xs).
Reformulating steepsum into a fold allows us to compute it
in one traversal.

• Not every function can be expressed as a fold. For
example, tail :: [a]→ [a] is not a fold!

Longest Prefix

• The function call takeWhile p xs returns the longest prefix
of xs that satisfies p:

takeWhile p [ ] = [ ]

takeWhile p (x : xs) =

if p x then x : takeWhile p xs
else [ ] .

• E.g. takeWhile (≤ 3) [1, 2, 3, 4, 5] = [1, 2, 3].
• It can be defined by a fold:

takeWhile p = foldr (tke p) [ ],
tke p x xs = if p x then x : xs else [ ].

• Its dual, dropWhile (≤ 3) [1, 2, 3, 4, 5] = [4, 5], is not a fold.

All Prefixes

• The function inits returns the list of all prefixes of the
input list:

inits [ ] = [[ ]],
inits (x : xs) = [ ] : map (x :) (inits xs).

• E.g. inits [1, 2, 3] = [[ ], [1], [1, 2], [1, 2, 3]].
• It can be defined by a fold:

inits = foldr ini [[ ]],
ini x xss = [ ] : map (x :) xss.

All Suffixes

• The function tails returns the list of all suffixes of the
input list:

tails [ ] = [[ ]],
tails (x : xs) = let (ys : yss) = tails xs

in (x : ys) : ys : yss.
• E.g. tails [1, 2, 3] = [[1, 2, 3], [2, 3], [3], [ ]].
• It can be defined by a fold:

tails = foldr til [[ ]],
til x (ys : yss) = (x : ys) : ys : yss.

144 / 214



Scan

• scanr f e = map (foldr f e) · tails.
• E.g.

scanr (+) 0 [1, 2, 3]
= map sum (tails [1, 2, 3])
= map sum [[1, 2, 3], [2, 3], [3], [ ]]
= [6, 5, 3, 0].

• Of course, it is slow to actually perform map (foldr f e)
separately. By fold-fusion, we get a faster implementation:

scanr f e = foldr (sc f) [e],
sc f x (y : ys) = f x y : y : ys.

• Folds are a specialised form of induction.
• Inductive datatypes: types on which you can perform
induction.

• Every inductive datatype give rise to its fold.
• In fact, an inductive type can be defined by its fold.

Fold on Natural Numbers

• Recall the definition:

data Nat = 0 | 1+ Nat .

• Constructors: 0 :: Nat, (1+) :: Nat→ Nat.
• What is the fold on Nat?

foldN ::

(a→ a)→ a

→ Nat→ a

foldN f e 0 = e
foldN f e (1+ n) = f (foldN f e n) .

Examples of foldN

• (+n) = foldN (1+) n

.

0+ n = n
(1+ m) + n = 1+ (m+ n) .

• (×n) = foldN (n+) 0

.

0× n = 0
(1+ m)× n = n+ (m× n) .

• even = foldN not True

.

even 0 = True
even (1+ n) = not (even n) .

Fold-Fusion for Natural Numbers

Theorem (foldN-Fusion)
Given f :: a→ a, e :: a, h :: a→ b, and g :: b→ b, we have:

h · foldN f e = foldN g (h e) ,

if h (f x) = g (h x) for all x.

Exercise: fuse even into (+)?

Folds on Trees

• Recall some datatypes for trees:

data ITree a = Null | Node α (ITree a) (ITree a) ,
data ETree a = Tip a | Bin (ETree a) (ETree a) .

• The fold for ITree, for example, is defined by:

foldIT ::

(a→ b→ b→ b)→ b→

ITree a→ b

foldIT f e Null = e
foldIT f e (Node a t u) = f a (foldIT f e t) (foldIT f e u) .

• The fold for ETree, is given by:

foldET ::

(b→ b→ b)→ (a→ b)→

ETree a→ b

foldET f g (Tip x) = g x
foldET f g (Bin t u) = f (foldET f g t) (foldET f g u) .

Some Simple Functions on Trees

• To compute the size of an ITree:

sizeITree = foldIT (λx m n→ 1+ (m+ n)) 0 .

• To sum up labels in an ETree:

sumETree = foldET (+) id.

• To compute a list of all labels in an ITree and an ETree:

flattenIT =foldIT (λx xs ys→ xs++[x] ++ ys) [ ],
flattenET =foldET (++) (λx→ [x]).

• Exercise: what are the fusion theorems for foldIT and
foldET?

Maximum Segment Sum

Finally we have introduced enough concepts to tackle the
maximum segment sum problem!

Maximum Segment Sum: given a list of numbers, find the
maximum possible sum of a consecutive segment.

Can be traced to 1984 in Dijkstra and Feijen’s Een methode van
programmeren,

Probably made famous by Bentley, and became a pet topic of
the program derivation community after being given a formal
treatment by Gries.

Perhaps the most popular example in program derivation. The
calculation we present here is close to that of Gibbons.

Specifying Maximum Segment Sum

• A segment can be seen as a prefix of a suffix.
• The function segs computes the list of all the segments.

segs = concat ·map inits · tails.

• Therefore, mss is specified by:

mss = max ·map sum · segs.

The Derivation!

We reason:

max ·map sum · concat ·map inits · tails

= { since map f · concat = concat ·map (map f) }
max · concat ·map (map sum) ·map inits · tails

= { since max · concat = max ·map max }
max ·map max ·map (map sum) ·map inits · tails

= { since map f ·map g = map (f.g) }
max ·map (max ·map sum · inits) · tails .

Recall the definition scanr f e = map (foldr f e) · tails. If we can
transform max ·map sum · inits into a fold, we can turn the
algorithm into a scanr, which has a faster implementation.

Maximum Prefix Sum

Concentrate on max ·map sum · inits:

max ·map sum · inits
= { definition of init, ini x xss = [ ] : map (x :) xss }

max ·map sum · foldr ini [[ ]]

= { fold fusion, see below }
max · foldr zplus [0] .

The fold fusion works because:

map sum (ini x xss)
= map sum ([ ] : map (x :) xss)
= 0 : map (sum · (x :)) xss
= 0 : map (x+) (map sum xss) .

Define zplus x yss = 0 : map (x+) yss.

Maximum Prefix Sum, 2nd Fold Fusion

Concentrate on max ·map sum · inits:

max ·map sum · inits
= { definition of init, ini x xss = [ ] : map (x :) xss }

max ·map sum · foldr ini [[ ]]
= { fold fusion, zplus x xss = 0 : map (x+) xss }

max · foldr zplus [0]
= { fold fusion, let zmax x y = 0 ↑ (x+ y) }

foldr zmax 0 .

The fold fusion works because ↑ distributes into (+):

max (0 : map (x+) xs)
=0 ↑ max (map (x+) xs)
=0 ↑ (x+max xs) .

Back to Maximum Segment Sum

We reason:

max ·map sum · concat ·map inits · tails
= { since map f · concat = concat ·map (map f) }

max · concat ·map (map sum) ·map inits · tails
= { since max · concat = max ·map max }

max ·map max ·map (map sum) ·map inits · tails
= { since map f ·map g = map (f.g) }

max ·map (max ·map sum · inits) · tails

= { reasoning in the previous slides }
max ·map (foldr zmax 0) · tails

= { introducing scanr }
max · scanr zmax 0 .

Maximum Segment Sum in Linear Time!

• We have derived mss = max · scanr zmax 0, where
zmax x y = 0 ↑ (x+ y).

• The algorithm runs in linear time, but takes linear space.
• A tupling transformation eliminates the need for linear
space.

mss = fst ·maxhd · scanr zmax 0

where maxhd xs = (max xs,head xs). We omit this last
step in the lecture.

• The final program is mss = fst · foldr step (0, 0), where
step x (m, y) = ((0 ↑ (x+ y)) ↑ m, 0 ↑ (x+ y)).

A Quick Note on Type Classes



The Type of take

• Recall the definition:
take 0 xs = [ ]

take (1+ n) [ ] = [ ]

take (1+ n) (x : xs) = x : take n xs .
• The first argument has to be of a numeric type (e.g. Int),
since we pattern matched it against 0 and 1+.

• The second argument must be a list, since we patten
matched it against [ ] and (:).

• But the element of the list is not examined at all. It is
merely copied to the output.

The Type of take

• The type of take can be
• Int→ List Int→ List Int;
• Int→ List Char→ List Char, etc.

• There is a most general type: Int→ List a→ List a.
• The small letter means that a is a type variable. One can
imagine that there is an implicit ∀ that quantifies all type
varaibles: ∀a.Int→ List a→ List a.

The Identity Function

• For a more obvious example, consider the (simple but
important) identity function

id x = x .

• The argument is not touched at all.
• It may have type Int→ Int, Char→ Char, or even
(Int→ Int)→ (Int→ Int).

• The most general type is a→ a.

Filter

• Recall filter:

filter :: (a→ Bool)→ List a→ List a
filter p [ ] = [ ]

filter p (x : xs) | p x = x : filter p xs
| otherwise = filter p xs .

• Still, in filter p (x : xs) we merely passes x to p, without
looking into x.

• Therefore filter works for any type a for which there exists
functions of type a→ Bool — which is true for all type a.

Counting Lowercase Characters

• For a counterexample, consider the following function:

lowers :: List Char→ Int
lowers [ ] = 0
lowers (x : xs) = if isLower x

then 1+ lowers xs
else lowers xs .

• The function counts the number of lowercase characters
in a string.

• It is equivalent to length · filter isLower.
• x is passed to isLower, which forces x to be a Char.

Parametric Polymorphism

• Polymorphism: allowing a piece of code to have many
types, such that it can be used in many occasions.

• Indeed, take can be applied to all types of lists. We do not
need to define a separate version for List Int,
List (Int→ Int).

• Parametric polymorphic, as we have seen just now, is
common in many functional programming languages.

• When take n :: List a→ List a is applied to an argument,
say [1, 2, 3], the type variable a is instantiated to the type
of the argument (Int in this case).

• The type variable a behaves like a parameter, thus the
name.

• Observe: the same piece of code (e.g. take, filter) works for
all instantiations of a .

• Object-oriented languages often adopt another kind of
polymorphism for operator overloading, called ad-hoc
polymorphism, to be introduced later.

Membership Test

• Given the definition below, elem x xs yields True iff. x
occurs in xs.

elem x [ ] = False
elem x (y : xs) | x = = y = True

| otherwise = elem x xs .
• It could have type Int→ List Int→ Bool,
Char→ List Char→ Bool, etc.

• We do not want to define elem once for each type, thus we
wish that it has a polymorphic type, say a→ List a→ Bool.

• However, not all values can be tested for equality! The
operator (= =) is defined for some types, but not all types.
For example, we cannot in general decide whether two
functions are equal.

• Thus elem cannot have type, for example,
(Int→ Int)→ List (Int→ Int)→ Bool.

The Eq Class

• There is such a definition in the Standard Prelude:
class Eq a where

(= =) :: a→ a→ Bool .

• which says that a type a is in the type class Eq if there is
an operator (= =), of type a→ a→ Bool, defined.

• Int is in Eq since we can define (= =) for numbers. So is
Char, although (= =) for Char implements a different
algorithm from that of Int.

Type of elem

• The most general type of elem is
Eq a⇒ a→ List a→ Bool,

• which means that elem takes a value of type a and a list of
type List a and returns a Bool, provided that a is in Eq.

• The additional constraint arises from the fact that elem
calls (= =).

Instance Declaration

• To use elem on concrete types, we have to teach Haskell
how to check equality for each type. The following are
defined somewhere in the Haskell Prelude:

instance Eq Int where
m = = n = {- how to check equality for Int -}

instance Eq Char where
m = = n = {- how to check equality for Char -}

• It is not possible to give a definition for, for example
Eq (a→ a). Thus elem cannot be applied to such types.

Instance Declaration

• When we define a new type, we might want to teach
Haskell how to check equality:

data Color = Red | Green | Blue . . .

•

instance Eq Color where
Red = = Red = True
Red = = Green = False

. . .

Summary So Far…

• Class declaration:
class Eq a where

(= =) :: a→ a→ Bool .
• The method (= =) then has type Eq a⇒ a→ a→ Bool.

• Instance declaration:
instance Eq MyType where
x = = y = . . .

• (= =) above should have type MyType→ MyType→ Bool, but
the type is not written.

• A function that calls a function with class constraint Eq a
(e.g. (= =)) also has the constraint in its type:

elem :: Eq a⇒ a→ List a→ Bool
elem = . . . = = . . .

• elem 2 [1, 2, 3] is allowed because there is an instance
declaration for Eq Int, while elem id [id, (1+), (2+)] is not
(unless you define and instance Eq (Int→ Int)).

Ad-hoc Polymorphism

• Note that (= =) for Int is a different program from that for
Char.

• Type classes is thus a way to describe operator loading —
using one name to refer to different piece of code.

• Such mechanisms are often called ad-hoc polymorphism.
• Compare with parametric polymorphism, where the same
code, say, the same definition of take, works for all types.

Other Important Type Classes

• Show: things that can be printed (converted to string).
• Read: things that can be parsed from strings.
• Num: things that behave like numbers (with addition,
multiplication, etc).

• Integral: things that behave like integers.
• Monad, Functor…hope we will be able to talk about them
later!

• Use :i in GHCi to find out what methods and instances
each class has!

Derived Instances

• The Haskell compiler may automatically construct some
routine instance declarations, to save you some typing.
E.g.

data Colors = Red | Green | Blue
deriving (Eq, Show,Read) .

Instance Inheritance

• How do we check whether two lists are equal? We can do
so if we know how to check whether their elements are
equal.

instance Eq a⇒ Eq (List a) where
[ ] = = [ ] = True
[ ] = = (x : xs) = False
(x : xs) = = [ ] = False
(x : xs) = = (y : ys) = x = = y ∧ xs = = ys .

• Note that in x = = y, the (= =) refers to the method for type a,
while the (= =) in xs = = ys is a recursive call.

Instance Inheritance

• Another example:

instance (Eq a, Eq b)⇒ Eq (a,b) where
(x1, y2) = = (x2, y2) = (x1 = = x2) ∧ (y1 = = y2) .

• All the three (= =) in the expression above refer to different
methods!

176 / 214



The Type Class Ord

• Another type class Ord includes things are can be
“ordered”:

class Eq a⇒ Ord a where
(<) :: a→ a→ Bool
(≥) :: a→ a→ Bool
(>) :: a→ a→ Bool
(≤) :: a→ a→ Bool . . .

• The declaration Eq a⇒ Ord a intends to mean that for a
type a to be in class Ord it has to be in class Eq.

• The methods (<), (≥), etc, is allowed to use (= =).
• Logically, it makes more sense to write Eq a⇐ Ord a . But
it’s a historical mistake that has been made.

• The function sort that sorts a list might have type
Ord a⇒ List a→ List a.

Class Hierarchy

• Inheritance between type classes are not to be confused
with inheritance between types.

• Through inheritance, type classes form a hierarchy.
• Types in the standard Haskell Prelude form a complex
hierarchy.

• Other libraries may extend the existing hierarchy or build
their own hierarchy.

Standard Haskell Numerical Classes

Notes

• The name “type class” is merely a mechanism for operator
loading and shall not be confused with classes in object
oriented languages.

• Type classes are an important feature of Haskell. Use of
type classes has extended far beyond the inventors had
imagined.

Monads and Effects



Side Effects

It is a misconception that functional languages do not allow
side effects. In fact, many of them allow a variety of effects.

It is just that side effects must be introduced in a disciplined
manner.

Disciplined? Such that we can use side effects, and still be
able to reason about programs.

Side effects: anything a function does other than returning a
value.

• reading/writing to a variable,
• raising an exception,
• input/output,
• partialty (possible failure),
• non-determinism,
• non-termination... and many more.

Impurity Made Explicit in Type

Impure programs (programs in which a side effect may incur)
and pure programs are separated by type.

An expression of type m a denotes a computation that, if run,
may yield a result of type a. During its execution, some side
effects may incur.

Such expressions are built using operators that are expected
to satisfy certain, agreed laws.

Programmers, Laws, Implementation

Programmers use operators to build programs. These
operators are supposed to satisfy a set of agreed laws.

The laws specify behaviours of these operators, with which the
programmers can reason about their programs.

Library implementors implement these operators, and ensure
that they do satisfy these laws.

The laws form an interface between the programmers and the
library implementators.

Monads

One of the ways to structure effectful programs is through
monads.

Two operators:

class Monad m where
return :: a→ m a
(>>=) :: m a→ (a→ m b)→ m b .

return x denotes a program that simply returns x, with no side
effects incurred.

m>>= f is a program that, when run, executes m, sends the
result to f, which generates a program, and runs the resulting
program.

A Bland Program

return (3+ 4) >>=λx→
return (x× x) >>=λy→
return (y+ 1) .

This is a simple program that is expected to return 50. It may
have type Monad m⇒ m Int.

Hmmm... not very impressive. Isn’t that just function
application written the other way round?

That is because we have not introduce any effectful operators
yet.

Monad Laws

Operators return and (>>=) should satisfy the three monad
laws:

left unit : return x>>= f = f x
right unit : f>>= return = f
associativity : (m>>= f)>>= g = m>>= (λx→ f x>>= g)

Programs That Can Fail

Define
class Monad m⇒ MonadFail m where
fail :: m a .

where fail denotes failure.

The only law we demand is

fail>>= f = fail .

On paper we sometimes write fail as /0.

Programs That Can Fail

Define
class Monad m⇒ MonadFail m where
fail :: m a .

where fail denotes failure.
The only law we demand is

fail>>= f = fail .

On paper we sometimes write fail as /0.

188 / 214



Exception Catching

Extending from MonadFail:

class MonadFail m⇒ MonadExcept m where
catch :: m a→ m a→ m a .

Laws: catch and /0 form a monoid:

catch /0 h = h ,
catch m /0 = m ,

catch m (catch h h′) = catch (catch m h) h′ ,

and unexceptional computations needs no handler:

catch (return x) h = return x .

Shortcut Product

Recall that product computes the product of all numbers in a
given list:

product :: List Int→ Int
product [ ] = 1
product (x : xs) = x× product xs ,

But, if we know in advance that there is a 0 in the list, we can
just return 0, right?

scutprod :: MonadExcept m⇒ List Int→ m Int
scutprod xs =

catch (if elem 0 xs then fail
else return (product xs))

(return 0) .

Show that scutprod xs = return (product xs).

Non-Deterministic Monad

Another possible extension of MonadFail:

class MonadFail m⇒ MonadNondet m where
([]) :: m a→ m a→ m a .

Laws: ([]) and /0 form a monoid:

/0 []m = m ,
m [] /0 = m ,

m [] (n [] k) = (m [] n) [] k ,

and a left distributivity law:

(m [] n)>>= f = (m>>= f) [] (n>>= f) .

Note: the class hierarchy of monads in this course is different
from that of the standard Haskell library, but adapted from
Gibbons and Hinze, which I find more suitable for teaching.

Insertion Into a List

The program insert x ys non-deterministically inserts x into
one arbitrary position of ys.

insert :: MonadNondet m⇒ a→ List a→ m (List a)
insert x [ ] = return [x]
insert x (y : ys) = return (x : y : ys)[]

insert x ys>>= (return · (y :)) .

Prove:
insert x ys>>= (return ·map f) =
insert (f x) (map f ys) .

Maybe as Exception

A natural choice for implementing MonadFail is Haskell’s
standard Maybe type:

data Maybe a = Nothing | Just a ,

such that Nothing denotes failure and Just x denotes a
computation yielding result x.

instance Monad Maybe where
return = Just
Just x >>= k = k x
Nothing >>= k = Nothing .

instance MonadFail Maybe where
fail = Nothing ,

instance MonadExcept Maybe where
catch Nothing h = h
catch (Just x) h = Just x .

The List Monad

Meanwhile, List also forms a monad.

instance Monad List where
return x = [x]
xs>>= k = concat (map k xs) .

Empty List as Failure

One may denote failed computation by the empty list, and a
succeeded computation by a singleton list.

instance MonadFail List where
fail = [] ,

instance MonadExcept List where
catch [] h = h
catch xs h = xs .

Lists Representing Non-determinism

In fact, lists are often used to represent non-determinism.

instance MonadNondet List where
([]) = (++) .

Therefore, a non-deterministic computation, when
implemented as lists, actually gives you the list of all its
results.

Is this a faithful representation? That depends on what you
expect from non-determinism. For now, it does satisfy the laws
we expect. See Kiselyov for discussion on suitable
implementations of non-deterministic monads.

Maybe as Non-determinism?

“If I do not need all results, but only one, can I implement
MonadNondet using Maybe?”

Try a possible implemenation:

instance MonadNondet Maybe where
Nothing []m = m
Just x []m = Just x ,

which is... exactly like catch. That is usually a sign that
something is wrong.

Verify: does this implementation satisfy the left-distributivity
law?

List Monad for Backtracking

Lesson: even if we want only the first result, we still need to
backtrack and compute the next result when the current first
result fails.

This is in fact what happens in the List monad. With lazy
evaluation we do not actually compute a list of results, but
keeping enough information to backtrack. This is how
backtracking is often represented in Haskell.

A Single, Unnamed Variable

For the State effect we introduce two operators that
respectively reads and writes to an unnamed variable:6

class Monad m⇒ MonadState st m where
get :: m st
put :: st→ m () .

6Our naming convention here: st is the type of the state, while s, s0, etc. are
values whose type could be st.

Laws for State Effect

We usually assume the following rules:

get-put : get>>= put = return () ,
put-get : put s>> get = put s>> return s ,
put-put : put s>> put s′ = put s′ ,
get-get : get>>= λs→ get>>= λs′ → f s s′ =

get>>= λs→ f s s ,

where m>> n = m>>= λs→ n is a shorthand we often use
when n does not need the result from m.

Summing Up a List using a State

Well, this is perhaps among your first few imperative programs.

loop [ ] = get
loop (x : xs) = get>>= λs→

put (s+ x)>> loop xs ,

with which you can sum up a list by put0>> loop xs.
Does it compute sum xs?

203 / 214



Loop and Fold-Left

Well, not quite... if (+) were not associative. What loop actually
computes is a foldl. Recall

foldl :: (b→ a→ b)→ b→ List a→ b
foldl (⊕) s [ ] = s
foldl (⊕) s (x : xs) = foldl (⊕) (s⊕ x) xs .

The do-Notation

Note: Haskell support a syntax called the do-notation which,
informally, allows m>>= x→ n to be written as n← m. The
function loop can be written as:

loop (⊕) [ ] = get
loop (⊕) (x : xs) = do s← get

put (s⊕ x)
loop (⊕) xs ,

which looks more like an imperative program.
In this lecture I prefer to make (>>=) explicit, to facilitate
reasoning. Nevertheless the do-notation is popular among
programmers.

Implementing MonadState

A program that has access to a state can be seen as a function
that maps an initial state to a pair of the returned value and
the final state:7

newtype StFun st a = MkS (st→ (a, st)) .

Given a computation having type StFun st a and an initial
state, we may execute it by:

run :: StFun st a→ st→ a
run (MkS k) s0 = k s0 .

7The same monad is called State in Haskell’s library. We use a different
name to avoid confusion.

State Operations

For all st, StFun st is a monad:
instance Monad (StFun st) where
return x = MkS (λs→ (x, s))
MkS f>>= k = MkS (λs→ let (x, s′) = f s

MkS h = k x
in h s′) ,

and it is a state monad:
instance MonadState st (StFun st) where
get = MkS (λs→ (s, s))
put s′ = MkS (λs→ ((), s′)) .

207 / 214



Destructive Update

This monad StFun s might not be what you expect: it simulates
state passing and updating, but does not actually update any
memory cell.

Currently in the Haskell library, an actual destructive update
can be performed in two monads: ST and IO.

The ST Monad

The monad ST is more complex than the state effect we have
discussed. Instead of get and put, there are methods for
creating new references, new arrays, accessing and updating
variables and arrays, etc.

In the type ST s a, the type variable s no longer stands for the
type of the state. It is never instantiated. Instead it is used to
ensure that the state cannot be leaked outside the monadic
computation. See Launchbury and Peyton Jones for details.

It is possible to wrap ST with an interface such that it can work
like MonadState. It is rarely done, though.

The IO Monad

The IO monad is where the programmer has access to all
unsafe features: destructive update, reading, printing, file
access, multi-threading using multable variables...

There is no way to privately “run” a computation of type IO. It
can only be run by the system, in the topmost level.

ST can be converted to IO.

State and Non-Determinism

Can a monad be in both MonadNondet and MonadState s?

It must provide operators from both effects: fail, ([]), get, put,
satisfy all the laws, and perhaps some additional laws stating
how the operators of different effects interact.

Local State

One possibility:

newtype StL st a = MkSL (st→ List (a, st)) .

Each non-deterministic branch has its own final state.
How to define its Monad, MonadNondet, and MonadState s
instance methods?

Global State

One might imagine a monad where all non-deterministic
branches share one global state:

newtype StG st a = MkSG (st→ (List a, st)) .

But no, this is not even a monad — the associativity of (>>=)
fails to hold.

It is tricky designing monads involving multiple effects.

Yet, a real program may use several effects — exceptions,
multiple states, input/output...

Modular Construction of Monads

Question: given a collection of desired effects, can we
construct, in a modular manner, a monad supporting these
effects?

Monad transformer is a popular approach. Effect handing is a
recent new alternative.

Question: how to ensure that the monad so constructed obey
the laws we specify?

This is still a research topic.

